** CODE for replication of "Interpersonal Resources and Insider/Outsider Dynamics in Party Office "
** Authors: Javier Martínez-Cantó and Tània Verge
** Journal: Comparative Political Studies (CPS)

** Before running this code, please see the "readme_replication" file for general instructions. 
** All procedures and analysis were executed using STATA 14.1

**********************************************************
*** Analysis 2: Unequal allocation of positional power ***
**********************************************************

*** Importing the dataset
clear 

use "data_necs.dta" // This dataset contains yearly observations for each NEC member

*** Installing required packages (if necesary)
*ssc install estout, replace
*ssc install blindschemes, replace
*ssc install coefplot, replace
*net install grc1leg, from(http://www.stata.com/users/vwiggins) replace

* set scheme plotplain

*** Keep only one observation per each parson in each party conference.
keep if years_since_last_partyconference == 0
drop if partyconference_type==1

***************************************************************
*** Analysis 2 - Descriptive analysis: Figure 1 (Main text) ***
***************************************************************
* Total
graph bar  , over(gender) over(period_5y) /// 
	asyvars percentages stack bar(1, fcolor(gs5)) bar(2, fcolor(gs16)) legend(order(1 "Men" 2 "Women") rows(1) position(6)) title("(a) All seats") ytitle("Percentage") name(keyplot1, replace)  fxsize(300) fysize(300)
* Men	
graph bar if party_hardcore==1 , over(gender) over(period_5y) /// 
	asyvars percentages stack bar(1, fcolor(gs5)) bar(2, fcolor(gs16)) legend(order(1 "Men" 2 "Women") rows(1) position(6)) title("(b) Key posts") ytitle("Percentage") name(keyplot2, replace)  fxsize(300) fysize(300)
* Women	
graph bar if party_hardcore==0 , over(gender) over(period_5y) /// 
	asyvars percentages stack bar(1, fcolor(gs5)) bar(2, fcolor(gs16)) legend(order(1 "Men" 2 "Women") rows(1) position(6)) title("(c) Non-key posts") ytitle("Percentage") name(keyplot3, replace)  fxsize(300) fysize(300)
* Together
grc1leg keyplot1 keyplot2 keyplot3, hole(2)
graph export "descriptive_key_positions.png", replace
	

*******************************************
*** Analysis 2 - Multivariate analysis  ***
*******************************************

* Globals capturing different sets of variables
global personal 							age university_degree isei 	
global family	 							childless	
global political_offices 					seniority_log national_politics_experience_log  
global inter_personal 						years_as_party_member_log youth_org leader_coincidence_log

global controls 							nec_seniority previous_key_post gender_internal_rules share_female_nec total_nec nec_election_mode
global party 								i.party 

global personal_ 							c.age c.university_degree c.isei
global family_	 							c.childless	
global inter_personal_						c.years_as_party_member_log c.youth_org c.leader_coincidence_log 
global political_offices_ 					c.seniority_log c.national_politics_experience_log  
global controls_ 							c.nec_seniority c.previous_key_post c.gender_internal_rules c.share_female_nec c.total_nec c.nec_election_mode
global party_ 								i.party 

* Remove observations with missing data
drop if full==0

* Full sample, regression models
logistic party_hardcore i.gender 																						$controls $party_, vce(cluster party)
eststo portfolios1
logistic party_hardcore i.gender $personal 																			$controls $party_  , vce(cluster party)
eststo portfolios2
logistic party_hardcore i.gender $political_offices 																	$controls $party_ , vce(cluster party)
eststo portfolios3
logistic party_hardcore i.gender $inter_personal 																		$controls $party_ , vce(cluster party)
eststo portfolios4
logistic party_hardcore i.gender $personal $political_offices $inter_personal  										$controls $party_, vce(cluster party)
eststo portfolios5
logistic party_hardcore i.gender##($personal_ $political_offices_ $inter_personal_ 									$controls_ $party_) , vce(cluster party)
eststo portfolios6

* Sub-sample, regression models
logistic party_hardcore i.gender 																						$controls $party_ if party<3402, vce(cluster party)
eststo portfolios7
logistic party_hardcore i.gender $personal 																			$controls $party_ if party<3402, vce(cluster party)
eststo portfolios8
logistic party_hardcore i.gender $family 																				$controls $party_ if party<3402, vce(cluster party)
eststo portfolios9
logistic party_hardcore i.gender $political_offices 																	$controls $party_ if party<3402, vce(cluster party)
eststo portfolios10
logistic party_hardcore i.gender $inter_personal 																		$controls $party_ if party<3402, vce(cluster party)
eststo portfolios11
logistic party_hardcore i.gender $personal $family $political_offices_ $inter_personal_ 					 			$controls $party_ if party<3402, vce(cluster party)
eststo portfolios12
logistic party_hardcore i.gender##($personal_ $family_ $political_offices_ $inter_personal_ 							$controls_ $party_) if party<3402, vce(cluster party)
eststo portfolios13

* Tables
esttab  portfolios1 portfolios2 portfolios3 portfolios4 portfolios5 portfolios6  , ///
compress se  noomitted  interaction(" X ") dropped("Ref.") nogaps star(+ .10 * .05 ** .01 *** .001) ///	
label order(*.gender $personal $political_offices $inter_personal $controls ) indicate("Party FE = *.party")

esttab  portfolios7 portfolios8 portfolios9 portfolios10 portfolios11 portfolios12 portfolios13, ///
compress se  noomitted  interaction(" X ") dropped("Ref.") nogaps star(+ .10 * .05 ** .01 *** .001) ///	
label order(*.gender $personal $family $political_offices $inter_personal $controls ) indicate("Party FE = *.party")

esttab  portfolios5 portfolios6 portfolios12 portfolios13, ///
compress se  noomitted  interaction(" X ") dropped("Ref.") nogaps star(+ .10 * .05 ** .01 *** .001) ///	
label order(*.gender $personal $family $political_offices $inter_personal $controls ) indicate("Party FE = *.party") varwidth(30)

* Appendix D: Tables 13 to 15

** Table 13
esttab portfolios1 portfolios2 portfolios3 portfolios4 portfolios5 portfolios6  using "portfolios_allocation_regressions1.tex", replace ///
compress se  noomitted  interaction(" X ") dropped("Ref.") nogaps star(* .05 ** .01 *** .001) ///	
mtitle("Model 1" "Model 2" "Model 3" "Model 4" "Model 5" "Model 6" "Model 7" "Model 8" "Model 9")  /// 
label nonumbers booktabs longtable pr2 scalars("ll Log lik.")  alignment(D{.}{.}{-1}) indicate("Party FE = *.party")  /// 
addnote("\begin{minipage}[t]{0.9\textwidth} Clustered standard errors by political party. Models including the variable childless are calculated using only PP and PSOE observations. \end{minipage}") ///
title("Determinants of obtaining a key post in parties' NEC (Full sample)\label{table.portfolios_allocation_regressions1}") ///
order(*.gender $personal $political_offices $inter_personal $controls ) 

** Table 14
esttab portfolios7 portfolios8 portfolios9 portfolios10 portfolios11 portfolios12 portfolios13  using "portfolios_allocation_regressions2.tex", replace ///
compress se  noomitted  interaction(" X ") dropped("Ref.") nogaps star(* .05 ** .01 *** .001) ///	
mtitle("Model 1" "Model 2" "Model 3" "Model 4" "Model 5" "Model 6" "Model 7" "Model 8" "Model 9")  /// 
label nonumbers booktabs longtable pr2 scalars("ll Log lik.")  alignment(D{.}{.}{-1}) indicate("Party FE = *.party")  /// 
addnote("\begin{minipage}[t]{0.9\textwidth} Clustered standard errors by political party. Models including the variable childless are calculated using only PP and PSOE observations. \end{minipage}") ///
title("Determinants of obtaining a key post in parties' NEC (Sub-sample)\label{table.portfolios_allocation_regressions2}") ///
order(*.gender $personal $family $political_offices $inter_personal $controls ) 

** Table 15
esttab  portfolios5 portfolios6 portfolios12 portfolios13 using "portfolios_allocation_regressions3.tex", replace ///
compress se  noomitted  interaction(" X ") dropped("Ref.") nogaps star(* .05 ** .01 *** .001) ///	
mgroups("Full sample" "Sub-sample", pattern(0 0 1 1)) mtitle("Model 1" "Model 2" "Model 3" "Model 4" "Model 5" "Model 6" "Model 7" "Model 8" "Model 9")  ///
label nonumbers booktabs longtable pr2 scalars("ll Log lik.")  alignment(D{.}{.}{-1}) indicate("Party FE = *.party")  /// 
addnote("\begin{minipage}[t]{0.9\textwidth} Clustered standard errors by political party. Models including the variable childless are calculated using only PP and PSOE observations. \end{minipage}") ///
title("Heterogeneous effects by gender in the distribution of key posts among NEC members\label{table.portfolios_allocation_regressions3}") ///
order(*.gender $personal $family $political_offices $inter_personal $controls ) 

* Main text: Figure 2

* Full sample
** Baseline model
coefplot 	 	(portfolios5, pstyle(p1) msymbol(circle) mlcolor(black) mfcolor(black) msize(small)), title("(a.1) Baseline model") ///
				keep(?.gender $personal $family $political_offices $inter_personal ) 		///
			 	||,  xline(0, lcolor(black)) xtitle("Coefficient") scheme(plottig)    drop( _cons)   /// 
				order(1.gender $personal $family $political_offices $inter_personal $controls ) headings(1.gender = "{bf:Individual factors}" years_as_party_member_log = "{bf:Interpersonal resources}") ///
				name(portfolios1, replace) legend(off) ///
				coeflabels(, wrap(45) notick interaction(" X ")) rename(.gender##?.age=age)  ytick(1.5 2.5 3.5 4.5 5.5 6.5 7.5 8.5 9.5 10.5 11.5 12.5 , notick glstyle(grid)) fxsize(100) fysize(100) //
** Interaction model
coefplot 	 	(portfolios6, label("Men") pstyle(p1) msymbol(diamond) mlcolor(black) mfcolor(black) msize(small) keep( $personal $partisan_experience $political_offices $inter_personal $party_position $family )) 		///
				(portfolios6, label("Women") pstyle(p1) msymbol(triangle) mlcolor(gs8) mfcolor(gs8) msize(medsmall) ciopts(lcolor(gs8)) keep( 1.gender#?.age 1.gender#?.university_degree 1.gender#?.isei 1.gender#?.years_as_party_member_log 1.gender#?.youth_org 1.gender#?.seniority_log 1.gender#?.national_politics_experience_log 1.gender#?.leader_coincidence_log 1.gender#?.party_hardcore 1.gender#?.nato 1.gender#?.childless )) 		///
			 	||,  xline(0, lcolor(black)) xtitle("Coefficient") scheme(plottig)    drop( _cons) title("(a.2) Interaction model")  ///
				order( age 1.gender#?.age university_degree 1.gender#?.university_degree isei 1.gender#?.isei seniority_log 1.gender#?.seniority_log national_politics_experience_log 1.gender#?.national_politics_experience_log years_as_party_member_log 1.gender#?.years_as_party_member_log youth_org 1.gender#?.youth_org  leader_coincidence_log 1.gender#?.leader_coincidence_log party_hardcore 1.gender#?.party_hardcore nato 1.gender#?.nato childless 1.gender#?.childless ) ///
				headings(age = "{bf:Individual factors}" years_as_party_member_log = "{bf:Interpersonal resources}") ///
				name(portfolios2, replace) legend(rows(1) position(6)) fxsize(100) fysize(150) /// 
				coeflabels(	1.gender#?.age=" " ///
							1.gender#?.university_degree=" " ///
							1.gender#?.isei=" " ///
							1.gender#?.years_as_party_member_log=" " ///
							1.gender#?.youth_org=" " ///
							1.gender#?.seniority_log=" " ///
							1.gender#?.national_politics_experience_log=" " ///
							1.gender#?.leader_coincidence_log=" " ///
							1.gender#?.party_hardcore=" " ///
							1.gender#?.nato=" " ///
							1.gender#1.nec_election_mode="Appointed by the leader" ///
							1.gender#2.nec_election_mode="Elected in closed lists" ///							
							1.gender#3.nec_election_mode="Elected in open lists" ///							
							1.gender#?.gender_internal_rules="Female quota size (in prop)" ///
							1.gender#?.share_female_mps="Share of women in PPG" ///
							1.gender#?.nec_renewal2="Prop. of NEC renewed" ///
							1.gender#?.total_nec="NEC size" ///
							1.gender#?.left_wing="Left-wing party" ///
							1.gender#?.childless="Childless" ///
									, wrap(45) notick interaction(" X ")) rename(.gender##?.age=age)  ytick(1.5 3.5 5.5 7.5 9.5 11.5 13.5 15.5 17.5 19.5, notick glstyle(grid)) // 

* Sub-sample
** Baseline model
coefplot 	 	(portfolios12, pstyle(p1) msymbol(circle) mlcolor(black) mfcolor(black) msize(small)), title("(b.1) Baseline model") /// 
				keep(?.gender $personal $family $political_offices $inter_personal ) 		///
			 	||,  xline(0, lcolor(black)) xtitle("Coefficient") scheme(plottig)    drop( _cons)   /// 
				order(1.gender $personal $family $political_offices $inter_personal $controls ) headings(1.gender = "{bf:Individual factors}" years_as_party_member_log = "{bf:Interpersonal resources}") ///
				name(portfolios3, replace) legend(off) /// 
				coeflabels(, wrap(45) notick interaction(" X ")) rename(.gender##?.age=age)  ytick(1.5 2.5 3.5 4.5 5.5 6.5 7.5 8.5 9.5 10.5 11.5 12.5 , notick glstyle(grid)) fxsize(100) fysize(100) // 
** Interaction model
coefplot 	 	(portfolios13, label("Men") pstyle(p1) msymbol(diamond) mlcolor(black) mfcolor(black) msize(small) keep( $personal $family $political_offices $inter_personal )) ///
				(portfolios13, label("Women") pstyle(p1) msymbol(triangle) mlcolor(gs8) mfcolor(gs8) msize(medsmall) ciopts(lcolor(gs8)) keep( 1.gender#?.age 1.gender#?.university_degree 1.gender#?.isei 1.gender#?.childless 1.gender#?.years_as_party_member_log 1.gender#?.youth_org 1.gender#?.seniority_log 1.gender#?.national_politics_experience_log 1.gender#?.leader_coincidence_log 1.gender#?.nato 1.gender#?.childless )) 		///
			 	||,  xline(0, lcolor(black)) xtitle("Coefficient") scheme(plottig)    drop( _cons) title("(b.2) Interaction model")  /// 
				order( age 1.gender#?.age university_degree 1.gender#?.university_degree isei 1.gender#?.isei  childless 1.gender#?.childless seniority_log 1.gender#?.seniority_log national_politics_experience_log 1.gender#?.national_politics_experience_log years_as_party_member_log 1.gender#?.years_as_party_member_log youth_org 1.gender#?.youth_org leader_coincidence_log 1.gender#?.leader_coincidence_log nato 1.gender#?.nato ) ///
				headings(age = "{bf:Individual factors}" years_as_party_member_log = "{bf:Interpersonal resources}") ///
				name(portfolios4, replace) legend(rows(1) position(6))  fxsize(100) fysize(150) /// 
				coeflabels(	1.gender#?.age=" " ///
							1.gender#?.university_degree=" " ///
							1.gender#?.isei=" " ///
							1.gender#?.years_as_party_member_log=" " ///
							1.gender#?.youth_org=" " ///
							1.gender#?.seniority_log=" " ///
							1.gender#?.national_politics_experience_log=" " ///
							1.gender#?.leader_coincidence_log=" " ///
							1.gender#?.party_hardcore=" " ///
							1.gender#?.nato=" " ///
							1.gender#1.nec_election_mode=" " ///
							1.gender#2.nec_election_mode=" " ///							
							1.gender#3.nec_election_mode=" " ///							
							1.gender#?.gender_internal_rules=" " ///
							1.gender#?.share_female_mps=" " ///
							1.gender#?.nec_renewal2=" " ///
							1.gender#?.total_nec=" " ///
							1.gender#?.left_wing=" " ///
							1.gender#?.childless=" " ///
									, wrap(45) notick interaction(" X ")) rename(.gender##?.age=age)  ytick(1.5 3.5 5.5 7.5 9.5 11.5 13.5 15.5 17.5 19.5, notick glstyle(grid)) // 
** All plots together									
graph combine portfolios1 portfolios2 , cols(1) xsize(5) ysize(10) title("(a) Full sample:", size(small)) name(portfoliosa, replace) xcommon
graph combine portfolios3 portfolios4 , cols(1) xsize(5) ysize(10) title("(b) Sub-sample, PP & PSOE:", size(small)) name(portfoliosb, replace) xcommon
graph combine portfoliosa portfoliosb ,  cols(2) xsize(10) ysize(10) xcommon 
graph save "portfolios_allocation.gph", replace
graph export "portfolios_allocation.png", replace

** Appendix D, Table 11 and 12: Descriptive statistics for analysis 2

tabulate nec_election_mode, generate(selection_mode_)
label variable selection_mode_1 "Selection: Appointed by the leader"
label variable selection_mode_2 "Selection: Elected in closed lists"
label variable selection_mode_3 "Selection: Elected in open lists"
tabulate party, generate(party_)
label variable party_1 "Party: AP/PP"
label variable party_2 "Party: PSOE"
label variable party_3 "Party: IU"
label variable party_4 "Party: Podemos"
label variable party_5 "Party: C's"
label variable party_6 "Party: Vox"

* Full sample
eststo all: quietly estpost summarize ///
    gender $personal $political_offices $inter_personal  $controls party_1 party_2 party_3 party_4 party_5 party_6

esttab all, ///
cells("mean( fmt(3)) sd( fmt(3)) min( fmt(3)) max( fmt(3))") ///
label compress // 

esttab all using "portfolios_allocation_descriptive1.tex", replace ///
cells("mean( fmt(3)) sd( fmt(3)) min( fmt(3)) max( fmt(3))") ///
label compress title("Descriptive statistics.") mtitle("Full sample") //

* Sub-sample
eststo all: quietly estpost summarize ///
    gender $personal $family $political_offices $inter_personal  $controls ///
	party_1 party_2 if party<3402

esttab all, ///
cells("mean( fmt(3)) sd( fmt(3)) min( fmt(3)) max( fmt(3))") ///
label compress  //

esttab all using "portfolios_allocation_descriptive2.tex", replace ///
cells("mean( fmt(3)) sd( fmt(3)) min( fmt(3)) max( fmt(3))") ///
label compress title("Descriptive statistics, only PP \& PSOE subsample with family information.")  mtitle("Sub-sample") //


***************************************
*** Analysis 2 - Robustness checks  ***
***************************************

** Robustness 1 - Excluding ex-officio and appointed NEC members from the full sample (Appendix D, table 16)

* Regressions
logistic party_hardcore i.gender $personal $partisan_experience $political_offices $inter_personal 					$controls $party_ if nato==0 , vce(cluster party)
eststo robust1
logistic party_hardcore i.gender##($personal_ $partisan_experience_ $political_offices_ $inter_personal_ 				$controls_ $party_) if nato==0 , vce(cluster party)
eststo robust2

logistic party_hardcore i.gender $personal $partisan_experience $political_offices $inter_personal $family				$controls $party_ if party<3402 & nato==0 , vce(cluster party)
eststo robust3
logistic party_hardcore i.gender##($personal_ $partisan_experience_ $political_offices_ $inter_personal_ $family_		$controls_ $party_) if party<3402 & nato==0 , vce(cluster party)
eststo robust4

esttab  robust1 robust2 robust3 robust4 , ///
compress se  noomitted  interaction(" X ") dropped("Ref.") nogaps star(+ .10 * .05 ** .01 *** .001) ///	
label order(*.gender $personal $political_offices $inter_personal $family $controls ) 

* Tables
esttab  robust1 robust2 robust3 robust4 using "portfolios_allocation_robust1.tex", replace ///
compress se  noomitted  interaction(" X ") dropped("Ref.") nogaps star(* .05 ** .01 *** .001) ///	
mgroups("Full sample" "Sub-sample", pattern(0 0 1 1)) mtitle("Model 1" "Model 2" "Model 3" "Model 4" "Model 5" "Model 6" "Model 7" "Model 8" "Model 9")  ///
label nonumbers booktabs longtable pr2 scalars("ll Log lik.")  alignment(D{.}{.}{-1}) indicate("Party FE = *.party")  /// 
addnote("\begin{minipage}[t]{0.9\textwidth} Clustered standard errors by political party. Models including the variable childless are calculated using only PP and PSOE observations. \end{minipage}") ///
title("Allocation of key posts within parties' NECs, robustness check 1:  Excluding ex-officio and appointed NEC members.\label{table.portfolios_allocation_regressions3}") ///
order(*.gender $personal $political_offices $inter_personal $family $controls ) 

** Robustness 2 - Alternative operationalizations of male homosocial capital (Appendix D, table 17 and 18)

* Regressions: Same region as leader 
logistic party_hardcore i.gender $personal $partisan_experience $political_offices years_as_party_member_log youth_org leader_region_coincidence 					$controls $party_, vce(cluster party)
eststo robust_b1
logistic party_hardcore i.gender##($personal_ $partisan_experience_ $political_offices_ c.years_as_party_member_log c.youth_org c.leader_region_coincidence 				$controls_ $party_) , vce(cluster party)
eststo robust_b2

logistic party_hardcore i.gender $personal $partisan_experience $political_offices years_as_party_member_log youth_org leader_region_coincidence $family				$controls $party_ if party<3402, vce(cluster party)
eststo robust_b3
logistic party_hardcore i.gender##($personal_ $partisan_experience_ $political_offices_ c.years_as_party_member_log c.youth_org c.leader_region_coincidence $family_		$controls_ $party_) if party<3402, vce(cluster party)
eststo robust_b4

esttab  robust_b1 robust_b2 robust_b3 robust_b4 , ///
compress se  noomitted  interaction(" X ") dropped("Ref.") nogaps star(+ .10 * .05 ** .01 *** .001) ///	
label order(*.gender $personal $partisan_experience $political_offices $inter_personal $family $controls ) // 

* Regressions: Same biological cohort as leader (birth -/+5 years) 
logistic party_hardcore i.gender $personal $partisan_experience $political_offices years_as_party_member_log youth_org leader_birth_coincidence 					$controls $party_, vce(cluster party)
eststo robust_b5
logistic party_hardcore i.gender##($personal_ $partisan_experience_ $political_offices_ c.years_as_party_member_log c.youth_org c.leader_birth_coincidence 				$controls_ $party_) , vce(cluster party)
eststo robust_b6

logistic party_hardcore i.gender $personal $partisan_experience $political_offices years_as_party_member_log youth_org leader_birth_coincidence $family				$controls $party_ if party<3402, vce(cluster party)
eststo robust_b7
logistic party_hardcore i.gender##($personal_ $partisan_experience_ $political_offices_ c.years_as_party_member_log c.youth_org c.leader_birth_coincidence $family_		$controls_ $party_) if party<3402, vce(cluster party)
eststo robust_b8

esttab  robust_b5 robust_b6 robust_b7 robust_b8 , ///
compress se  noomitted  interaction(" X ") dropped("Ref.") nogaps star(+ .10 * .05 ** .01 *** .001) ///	
label  order(*.gender $personal $partisan_experience $political_offices $inter_personal $family $controls ) 

* Tables
esttab portfolios5 portfolios6 robust_b1 robust_b2 robust_b5 robust_b6 using "robust_analysis2_interpersonal_operatinalization_fs.tex" , replace ///
compress se  noomitted  interaction(" X ") dropped("Ref.") nogaps star(* .05 ** .01 *** .001) ///	
mtitle("Years (log)" "--" "Same region" "--" "Same generation" "--") ///	
label nonumbers booktabs longtable pr2 scalars("ll Log lik.")  alignment(D{.}{.}{-1}) indicate("Party FE = *.party")  /// 
addnote("\begin{minipage}[t]{0.9\textwidth} Clustered standard errors by political party. Models including the variable childless are calculated using only PP and PSOE observations. \end{minipage}") ///
title("Allocation of key posts within parties' NECs, robustness check 2: Alternative operationalizations of male homosocial capital (full sample).\label{table.portfolios_allocation_regressions4}") ///
order(*.gender $personal $partisan_experience $political_offices $inter_personal $controls ) 

esttab portfolios12 portfolios13 robust_b3 robust_b4 robust_b7 robust_b8 using "robust_analysis2_interpersonal_operatinalization_ss.tex" , replace ///
compress se  noomitted  interaction(" X ") dropped("Ref.") nogaps star(* .05 ** .01 *** .001) ///	
mtitle("Years (log)" "--" "Same region" "--" "Same generation" "--") ///	
label nonumbers booktabs longtable pr2 scalars("ll Log lik.")  alignment(D{.}{.}{-1}) indicate("Party FE = *.party")  /// 
addnote("\begin{minipage}[t]{0.9\textwidth} Clustered standard errors by political party. Models including the variable childless are calculated using only PP and PSOE observations. \end{minipage}") ///
title("Allocation of key posts within parties' NECs, robustness check 2: Alternative operationalizations of male homosocial capital (sub-sample).\label{table.portfolios_allocation_regressions5}") ///
order(*.gender $personal $partisan_experience $political_offices $inter_personal $controls ) 
